```
'************************************************************
'This program was designed to operate a garage door drive that
'has remote control inputs as well as station up, down, and stop
'buttons. This drive also has a pulse generator that is used to
'detect if the drive has stalled during a move up or down.All the
'inputs are 5 volts DC and the up/down enable is interfaced by
'relays.
'Ver 1.d
'JVC 12/2005
'
'Ver 1.e
'JVC 11/7/2010
'Changed the up & down door limit stiched from normally low to
'normally high. This was done becausw the switched have poor contacts
'and pulling low is easrier then going high. Also it is safer because
'if the wire falls off of the arm breaks the door will stop
'
'************************************************************

' {$STAMP BS1}
' {$PBASIC 1.0}

'************************************************************
' define variables
'************************************************************


SYMBOL        UP            = BIT0      'word 0
SYMBOL        DOWN          = BIT1      'word 0
SYMBOL        RUNNING       = BIT2      'word 0
SYMBOL        STOP          = BIT3      'word 0
SYMBOL        RPM           = W3        'Speed of the drive in RPM
SYMBOL        DWN_LS        = PIN4      'down limit switch
SYMBOL        UP_LS         = PIN3      'up limit switch
SYMBOL        UP_PB         = PIN0      'up push button
SYMBOL        DWN_PB        = PIN1      'down push botton
SYMBOL        STP_PB        = PIN2      'stop push button




'************************************************************
'Initialization:
'************************************************************
  DIRS= %01100000
  LOW 5                                 'Turn off up
  LOW 6                                 'Turn off on
  Stop=1                                'Set the stop biit
  RUNNING=0                             'Clear the running bit
  RPM=0                                 'Clear the RPM count
  UP=0                                   'Clear the UP bit
  DOWN=0                                'Clear the DOWN bit

'************************************************************
' Main Program Logic
'************************************************************


Main:
  IF UP_PB=1  AND UP_LS=1  AND STP_PB=0 THEN GO_UP      'up PB + not up LS + not stop
  IF DWN_PB=1 AND DWN_LS=1 AND STP_PB=0 THEN GO_DOWN    'down PB + not dwn LS + not stop


LOOP:
  'DEBUG CLS,UP_PB,DWN_PB,STP_PB,UP_LS,DWN_LS,running,up,down,Stop,RPM
  IF UP=1 AND UP_LS=0 THEN GO_STOP      'Check if the up LS is made
```

```
  IF DOWN=1 AND DWN_LS=0 THEN GO_STOP    'Check if the down LS is made
  IF STP_PB=1 THEN GO_STOP               'Check for the stop PB

  PULSIN 7,1,W2                         'Read the pulse generator
  LET W2=W2/100                         'Convert count to RPM
  LET W2=60000/W2
  W2=W2 MAX 590                         'don't let RPM go read over 590 RPM

  RPM=W2
  IF RPM <10 THEN GO_STOP              'if there are no pulses then stop. door it stopped

  IF RUNNING=0 THEN Main
  GOTO LOOP

'************************************************************
' Subroutines
'************************************************************

GO_UP:
    Stop=0                              'Clear the stop bit
    LOW 6                               'Clear the down output
    HIGH 5                              'Turn on the up output
    UP=1                                'Set the UP bit
    RUNNING = 1                         'Set the RUNNING bit
    GOTO LOOP


GO_DOWN:
    Stop=0                              'Clear the stop bit
    LOW 5                               'Clear the up output
    HIGH 6                              'Set the down output
    DOWN=1                              'Set the DOWN bit
    RUNNING=1                           'Set the RUNNING bit
    GOTO LOOP


GO_STOP:
    LOW 5                               'Clear the up output
    LOW 6                               'Clear the down output
    STOP=1                              'Set the stop bit
    RUNNING=0                           'Clear the RUNNING bit
    UP=0                                'Clear the UP bit
    DOWN=0                              'Clear the DOWN bit
    RPM=0                               'Set RPM to zero
    GOTO MAIN


  END
```